
Capítulo 1 

Introdução às Funções Financeiras do 
Excel 
NESTE CAPÍTULO 

 Introduzindo o conceito fundamental do valor do dinheiro no tempo 
 Usando as funções financeiras básicas do Excel VP, VF, NPER, PGTO, e TAXA 
 Convertendo taxas de juros efetiva e nominal  
 Calculando custos efetivos de empréstimos usando diferentes sistemas de cotações de taxas 
 Calculando os pagamentos acumulados de juros e principal usando as funções CUMIPMT e 

CUMPRINC 
 Matching different interest and pagamento frequencies 
 Entendendo as limitações das funções VP, VF, NPER, PGTO, TAXA, CUMIPMT e CUMPRINC 

 
 
 
  

É BEM CERTO que a maioria dos usos comuns do Excel é para realizar cálculos envolvendo dinheiro. A 
cada dia, as pessoas tomam centenas de milhares de decisões financeiras baseadas nos números que são 
calculados numa planilha. Estas decisões vão desde uma simples (Eu posso me permitir a compra de um 
carro novo?) até uma complexa (Adquirindo a Corporação  XYZ  resultará num fluxo de caixa  positivo nos 
próximos 18 meses?). Este é o primeiro de três capítulos que discute cálculos financeiros que você pode 
realizar com o auxílio do Excel. 

Funções Financeiras Básicas do Excel                                     
Este capítulo apresenta muitos exemplos que usam cinco funções financeiras básicas do Excel. 

A sintaxe para estas funções é mostrada aqui (argumentos em negrito são argumentos exigidos): 

 VP(taxa, nper, pgto, vf, tipo) 
 VF(taxa, nper, pgto, vp, tipo) 
 PGTO(taxa, nper, vp, vf, tipo)  
 TAXA(nper, pgto, vp, vf, tipo, estimativa) 
 NPER(taxa, pgto, vp, vf, tipo) 

 

Como você verá, estas funções são extremamente flexíveis, e são úteis para uma grande variedade de 
problemas. Para usar estas funções efetivamente, você precisará entender três conceitos básicos: 

 Marcar os fluxos de dinheiro como positivo ou negativo 
 O conceito básico do valor do dinheiro no tempo 
 O conceito de taxas de juros equivalentes 

Estes conceitos estão todos cobertos neste capítulo e serão colocados em uso mais adiante nos capítulos 
subseqüentes. 

Convenção de Marcação dos Fluxos de Dinheiro 
Observe o seu extrato bancário, e ficará patente que o dinheiro flui! 
Quando se trata de funções financeiras do Excel, é crítico que você entende como “sinalizar” os fluxos de 
caixa. Em outras palavras, você usará o sinal positivo ou negativo? 
 
 
 
 
 
 
 
 
 
 
 

Terminologia Básica 
 Valor Presente (VP): Esta é a quantia principal. Se você investir $5.000 num CD 

(Certificado de Depósito) bancário, esta quantia representa o principal, ou valor presente, 
do dinheiro que você investiu. Se você tomou emprestado $15.000 para compra de um 
carro, esta quantia representa o principal ou valor presente do empréstimo. Valor Presente 
pode ser positivo ou negativo. 

 Valor Futuro (VF): Este é o principal mais os juros. Se você investir $5.000 por cinco anos 
e ganhar 6% de juros anuais, você receberá $6.312,38 no final do prazo de cinco anos. A 
quantia é o valor futuro do seu investimento de $5.000. Se você conseguir um empréstimo 
de automóveis de três-anos no valor de $15.000 e pagar 7% de juros anuais, você pagará 
um total de $16.673,16. Esta quantia representa o principal mais os juros que você pagou. 
O Valor Futuro pode ser ou positivo ou negativo. 

 Pagamento (PGTO): Este é ou o principal ou principal mais os juros. Se você deposita 
$100 por mês numa conta de poupança, $100 é o pagamento. Se você tiver uma 
pagamento de $825 numa hipoteca, os $825 são constituídos do principal e juros. 

 Taxa de juros: Juro é uma porcentagem do principal, usualmente expressa numa base 
anual. Por exemplo, você poderia ganhar 5,5% de juros anuais num CD bancário. Ou sua 
hipoteca poderá ter uma taxa de juros de 7,75%. 

 Período: Este representa o momento quando os juros são pagos ou ganhos. Por exemplo, 
um CD bancário que paga juros trimestralmente ou um empréstimo de veículos que exige 
pagamentos mensais. 

 Prazo: Esta é a quantia de tempo de juros. Um CD 12-meses tem um prazo de um  ano. 
Uma hipoteca de 30-anos tem um prazo de 30 anos. 



Bertolo  Introdução às Funções Financeiras do Excel   

Página 4                                                                                                                                                              Finanças no Excel    

 
 
 
 
 
 
 
 
 
 
 
 
 
 
 
 
 
 

 

 

 

 

 

 

 

 

 

 

 

 

 

 

 

 

 
Para resolver problemas financeiros usando as funções financeiras básicas do Excel, você precisa executar 
dois passos preliminares: 

1. Determinar a perspectiva dos fluxos de caixa do credor. Por exemplo, num problema de capitalização 
simples, você a está observando da perspectiva do depositante ou do banco? Num problema de hipoteca, 
você será o devedor, ou o credor? Quando calcular o valor de uma série de pagamentos futuros, você é o 
comprador (o pagamento sai pelo direito de receber), ou você é o vendedor (recebendo um pagamento por 
conceder aquele direito)? 

2. Determine se qualquer particular valor presente, pagamento, ou valor futuro vem para você (sinal 
positivo), ou vai de você (sinal negativo). 

A Relação entre NPER, PGTO e TAXA 
O Excel não “sabe” nada a respeito dos diferentes períodos de tempo tais como meses, semanas, ou 
anos. Ele meramente conta-os e espera você rotulá-los apropriadamente e estar certo de que você não os 
misture. 

O diagrama associado representa o conceito do valor do dinheiro no tempo usado pelas funções do 
Excel VP, VF, PGTO, NPER e TAXA. As setas representam fluxos de dinheiro, e seu sentido (positivo 
ou negativo). Qualquer problema com solução consite de quatro variáveis conhecidas e uma variável 
incógnita. A variável incógnita é o nome da função, e as variáveis conhecidas representam os 
argumentos da função. 

O diagrama deve ser do saldo em termos de fluxos descontados ou acumulados, negativo e positivo. O 
conceito permite somente uma única taxa de juros, a qual deve ser a taxa efetiva para o período de 
tempo medido pelo NPER. Similarmente, somente um nível de pagamento é permitido, e este deve ser 
por um período de tempo medido por NPER. O argumento Tipo no conceito mostra se os pagamentos 
são antecipados ou postecipados. 

Se você puder preencher quatro das cinco variáveis, o Excel pode resolver o problema. Há uma 
exceção: Se os pagamentos são envolvidos, o Excel precisa saber quando os pagamentos ocorrem (isto 
é, o argumento Tipo). 

Quando você tem uma empresa negociando com estes dois pontos, você será capaz de usar as funções 
financeiras do Excel para criar fórmulas financeiras efetivas — e ser capaz de interpretar os resultados 
retornados pelas fórmulas. 

Geralmente, dinheiro que entra para você é assumido com sinal positivo. Dinheiro que sai de você é 
assumido ser negativo. Por exemplo, se um problema de valor presente retorna um valor negativo, significa 
que esta quantia é paga no período de tempo zero. Se ela é positiva, o dinheiro é recebido. Considere um 
exemplo de cálculo de pagamentos de hipoteccas. Se você é o tomador do empréstimo, o empréstimo “vem 
para você”, e os pagamentos calculados terão um sinal negativo (os quais indicam que você os pagou). 
Quando se calcula a taxa de juros sobre um empréstimo de hipoteca, você deve tomar cuidado to sign the 
empréstimo valorand os pagamentos apropriadamente. Caso contrário, o Excel assumirá que eles são todos 
numa direção e gerará um erro. Por exemplo, a fórmula deve mostrar 

#NUM!, a qual indica uma taxa de retorno infinitamente alta (tudo virá para você e nada será pago por isto). 

Funções Acumulação, Desconto e Amortização  
Esta seção contém vários exemplos que demonstram o uso das cinco funções básicas do Excel para resolver 
problemas de capitalização e amortização. Embora a tendência seja observar a amortização e capitalização 
como problemas separados, eles são essencialmente os mesmos. De fato, a única diferença é no sinal dos 
fluxos de caixa. 
Podemos classificar estes problemas em problemas simples e complexo. Em problemas simples, lidamos 
com somente duas das três variáveis de caixa (valor presente, pagamento, e valor futuro). Em problemas 
complexos, lidamos com todas as três. 
Apesar de classificarmos estes como problemas simples e complexos, o Excel ainda exige um valor para 
todas as três variáveis de caixa. Portanto, usamos zero para o elemento “faltante”. 

Problemas Capitalização Simples  
Esta seção contém sete exemplos que demonstram problemas de capitalização simples. 
 

Todos os exemplos nesta seção estão disponíveis nas Planilhas da Pasta Funções 
Financeiras do Excel que acompanha. 
 

EXEMPLO 1 

Aplicando $1.000 numa caderneta de poupança, quanto será o montante acumulado após dois anos, a juros 
de 6% por ano? 

A Figura 1-1 mostra este problema montado numa planilha. 

Figura 1-1: Calculando um valor futuro 
 
Função exigida: VF(taxa; nper;pgto;vp; tipo) 
Esta fórmula retorna $1.123,60: 

=VF(6%;2,0;-1000;0) 

As fórmulas exemplos neste capítulo usam valores severamente codificados para os 
argumentos da função. Os exemplos nas Planilhas da Pasta Funções Financeiras do Excel usam 
referências de células para os argumentos da função. 
 
Note que este problema é estabelecido da perspectiva do depositante. Portanto, o depósito inicial (o 
argumento vp) é negativo. Nenhum pagamento regular é feito, assim o argumento pgto é 0. Com nenhum 
pagamento, o argumento tipo é irrelevante. 

Quando entrando com dados numéricos como argumentos da função, assegure-se de não 
inserir separadores de milhares. Por exemplo, digite 1000, não 1.000. 
Dependendo de sua configuração regional, os separadores de milhares podem ser os mesmos 
caracteres dos separadores de argumentos. 
EXEMPLO 2 

Se $1.000 capitalizou para $2.000 em nove anos, qual foi a taxa de crescimento anual média? 

Função exigida: TAXA(nper;pgto;vp;vf; tipo;estimativa) 
Esta fórmula retorna 8,005974%: 

=TAXA(9;0;-1000;2000;0) 
Este exemplo é da perspectiva do depositante, assim o argumento vp é negativo e o argumento vf (um direito 
a receber) é positivo. Devido o prazo ser expresso em anos, taxa é a taxa efetiva ao ano. 

EXEMPLO 3 

Depositando $100.000 e ganhando 4% ao ano, quanto tempo levará para tornar-me um milionário? 

Função exigida: NPER(taxa; pgto; vp;vf; tipo) 
Esta fórmula retorna 58,71: 

=NPER(4%;0;-100000;1000000;0) 
Este exemplo é da perspectiva do depositante. Portanto, o argumento vp é negativo e o argumento vf (o 
direito a receber $1 milhão) é positivo. 
Devido à taxa ser cotada em termos efetivos anuais, o resultado está em anos. 

EXEMPLO 4 

Se eu tenho R$125.463,45 em minha conta e se ganhei 2% de juros ao mês durante 12 meses, qual foi o 
depósito original?  

Função exigida: VP(taxa;nper;pgto;vf;tipo) 
Esta fórmula retorna –R$98.927,07: 

=VP(2%;12;0;125463,45;0) 
Com nenhum pagamentos regular, o argumento pgto é 0 e o argumento tipo é irrelevante. 
Devido aos R$125.463,45 na conta ser um direito a receber, o argumento vf leva o sinal positivo e o valor 
presente calculado é negativo. 

EXEMPLO 5 

Se eu deposito $200 por mês (começando hoje) numa conta ganhando 1,5% por mês, quanto terei após três 
anos? 

Função exigida: VF(taxa;nper;pgto; vp; tipo) 
Esta fórmula retorna $8.172,96: 

=VF(1,5%;36;-200;0;1) 
Neste exemplo, o prazo é cotado em anos, mas os juros e os pagamentos são mensais. Isto exige um cálculo 
preliminar. A abordagem mais direta é converter anos para meses. Uma outra opção é converter a taxa de 
juros para uma taxa efetiva anual, e daí converter os $200 para uma quantia equivalente por ano. Isto 
produzirá o mesmo resultado, mas é uma abordagem excessivamente complicada. 
Note que pagamentos começam “hoje” e são, portanto, antecipados, Consequentemente, oargumento tipo é 
1. Nenhum saldo presente é estabelecido, então o argumento vp é 0. 

Em todos os exemplos precedentes, as questões podem ser reformuladas tais que os negativos tornem-se 
positivos, e os positivos tornem-se negativos. Portanto, o Examplo 1 pode ser reformulado como segue. 

EXEMPLO 6 

Tomando emprestado $2.000 por quatro anos a 5% de juros, quanto você terá de pagar de volta? 

Função exigida: VF(taxa; nper; pgto;vp;tipo) 
Esta fórmula retorna –$2.431,01: 

=VF(5%;4;0;2000;0) 
Aqui a questão é da perspectiva do tomador do empréstimo, e a fórmula foi modificada tal que o empréstimo 
tomado inicialmente (o argumento vp) seja positivo. Nenhum pagamento regular foi feito, assim o 
argumento pgto é 0. Com nenhum pagamento, o argumento tipo é irrelevante. 
Os Exemplos 2 até o 5 podem também serem reformulados assim: O depositante torna-se o tomador do 
empréstimo, e o tomador do empréstimo torna-se o depositante. 

EXEMPLO 7 

Se $1.000 capitalizou para $4.000 em nove anos, qual foi a taxa de crescimento anual média? 

Função exigida: TAXA(nper;pgto;vp;vf;tipo; estimativa) 
Esta fórmula retorna 16,652904%: 

=TAXA(9;0;-1000;4000;0) 
Este exemplo é da perspectiva do depositante. Portanto, o argumento vp é negativo e o argumento vf (um 
direito a receber) é positivo. Devido ao prazo ser expresso em anos, a taxa é a taxa efetiva por ano. Com 
nenhum pagamento regular, o argumento é 0 e o argumento tipo é irrelevante. 

Uma característica importante dos cálculos financeiros é que eles podem ser verificados 
para se estabelecer a precisão da resposta. Isto pode ser feito “fora da planilha” usando uma 
calculadora financeira, ou ele pode ser feito usando a fórmula subjacente ou outra função. 
Os passos seguintes demonstram um método para verificar o resultado de 16,652904% para este exemplo: 

1. Calcule quanto $1.000 accumula em nove anos a taxa calculada. Esta fórmula retorna $4.000: 

=VF(16,652904%;9;0;-1000;0) 
2. Calcule o valor presente de $4.000, descontado à taxa calculada por nove anos. A fórmula seguinte retorna 
–$1.000: 

=VP(16,652904%;9;0;4000;0) 
3. Calcule quanto tempo leva $1.000 para acumular a $4.000 à taxa calculada. A fórmula seguinte retorna 
oito: 

=NPER(16,652904%;0;-1000;4000;0) 
4. Calcule o resultado usando o seguinte fórmula, que retorna 16,652904%: 

=(4000/--1000)^(1/9)-1 
Uma técnica para verificação cruzada é comparar o cálculo verificado com os dados originais de tal maneira 
que o método produza um erro de 0. Em todas as verificações anteriores, subtraindo os dados originais do 
cálculo de verificação produz um erro zero. Se todos os cálculos forem verificados e os erros calculados 
desta forma, a soma de todos os erros numa planilha aproximará de zero. É improvável ser exatamente zero 
devido aos erros de arredondamento. 

Os exemplos nas Planilhas da Pasta Funções Financeiras do Excel contêm fórmulas de 
verificação de erros. 

Problemas Complexos de Capitalização  

Esta seção descreve quatro exemplos de problemas complexos de capitalização. Existem dois tipos de 
problemas complexos de capitalização: 

_ Problemas que não tem valores nulos para quaisquer dois dos parâmetros chaves (valor presente, 
pagamento, e valor futuro), e exige uma solução para o terceiro parâmetro. 

_ Problemas que tem entradas não nulas para todos os três parâmetros (valor presente, pagamento, e valor 
futuro), e exige uma solução ou para a TAXA ou NPER. 

 

Todos os exemplos nesta seção estão disponíveis nas Planilhas da Pasta Funções 
Financeiras do Excel, junto com a verificação cruzada para garantir sua precisão. 
 

EXEMPLO 8 

Com um saldo inicial de $25.500 e pagamentos de $1.500 por mês (no final de cada mês), quanto eu 
acumularei durante três anos se eu ganho 0,85% por mês? 

A Figura 11-2 mostra este exemplo, monte-o numa planilha. 

 
 
Figura 1-2: Calculando um valor futuro 
Função exigida: VF(taxa;nper; pgto; vp; tipo) 
Esta fórmula retorna $97.447,44: 

=VF(,85%;36;-1500;-25500;0) 
O sinal negativo para o argumento vp pode ser confuso, porque ele representa um saldo atual (um direito a 
receber). Entretanto, devido a estarmos olhando adiante no tempo, ele é tratado como um depósito. 
Pagamentos e taxas são cotados com base mensal; portanto, o prazo de três anos deve ser convertido para 
meses. O VF é retornado como positivo, o qual é um direito a receber. 

EXEMPLO 9 

Meu saldo bancário a seis anos atrás era $15.000, e eu adicionei $2.500 ao término de cada ano. O saldo 
atual é $50.000. Qual foi o meu retorno anual médio? 

Função exigida: TAXA(nper; pgto; vp; vf; tipo; estimativa) 
Esta fórmula retorna 12,049840%: 

=TAXA(6;-2500;-15000;50000;0;0) 

TAXA é uma função particularmente poderosa, devido à solução poder ser somente obtida 
por iteração. Só raramente é necessário inserir uma taxa estimativa como o sexto argumento 
opcional. Se omitido, O Excel fornece a estimativa default de 0. 
EXEMPLO 10 

Minha conta tinha um débito de $12.000 e eu deposito $1.000 no final de cada mês. Quanto tempo levará 
para eu me tornar um milionário se eu ganhar um retorno médio de 0.6% por mês? 

Função exigida: NPER(taxa, pgto, vp, vf, tipo) 
A fórmula seguinte retorna 390,01 meses: 

=NPER(0,8%;-500;15000;1000000;0) 
Note que a questão é formulada tal que o débito seja um depósito. Portanto, ele requer o sinal negativo para 
o argumento vp. 

Se o saldo negativo é visto como um empréstimo, o valor futuro será positivo. Num tal caso, dois cálculos 
seriam exigidos se a taxa saldo negativo não for igual à taxa de depósito. Primeiro calcularíamos o tempo 
levado para atingir saldo zero, e daí então calcularíamos o tempo para atingir $1 milhão. 

Usando taxas de 0,9% para o saldo negativo e 0.6% para os depósitos, esta fórmula retorna 463,90 meses: 

=NPER(0,9%;-500;15000;0;0)+NPER(0,6%;-500;0;1000000;0) 
EXEMPLO 11 

Deposito $1.000 por mês (no final de cada mês) pretendo fazer isto nos próximos dez anos. Se eu preciso 
acumular  $1.000.000, quanto deveria depositar agora se a conta recebe 0,5% ao mês? 

Função exigida: VP(taxa; nper; pgto;vf; tipo) 
Esta fórmula retorna -$459.559,28: 

=VP(0,5%;120;-1000;1000000;0) 
Precisamos converter anos para meses para garantir a adequação dos argumentos pgto, taxa e nper. 

Se trabalhar nos primeiros 11 exemplos, você deverá ter entendido o processo: 

1. Determinar as funções exigidas. 

2. Determinar os sinais das entradas pgto, vp, e vf. 

3. Garantir que os períodos da taxa, nper e pgto são os mesmos (ou convertê-los para torná-los os mesmos). 

4. Inserir os argumentos na ordem correta (preferencilamente usando células referências). 

5. Considere o significado da resposta. 

6. Determine que função, ou cálculos, é exigido para uma verificação cruzada. 

7. Certifique-se de que o erro se aproxime de zero. 

Problemas Simples de Descontos  

Você pode pensar o desconto como “o inverso da capitalização ou acumulação.” Ao invés de capitalizar um 
valor presente para um valor futuro, estaremos determinando o valor presente de uma quantia futura. 

Como com a capitalização, poderemos ter problemas que envolvam dois ou três dos valores monetários VP, 
VF, ou PGTO. Onde somente dois são envolvidos, nós chamaremos de desconto simples e quando todos os 
três estiverem envolvidos, chama-los-emos de desconto complexo. 
 

Todos os exemplos nesta seção estão disponíveis nas Planilhas da Pasta Funções 
Financeiras do Excel. Cada exemplo também contém uma verificação cruzada para garantir a 
precisão dos cálculos. 
EXEMPLO 12 

Qual é o valor presente do direito de receber $25.000 daqui cinco anos, descontado a 6,5% por ano? 

A Figura 11-3 mostra este exemplo, monte numa planilha. 

 
 
Figura 1-3: Calculando um valor presente 
Função exigida: VP(taxa; nper; pgto;vf;tipo) 
Esta fórmula retorna –$18.247,02: 

=VP(6.5%;5;0;25000;0) 
Note a lógica dos sinais. Se tivermos um direito a receber, o argumento vf é positivo — devemos pagar no 
presente para receber este direito positivo no futuro. 

Com nenhum pagamento, o argumento tipo é irrelevante. 

A precisão dos cálculos pode ser assegurada pela verificação cruzada da resposta com outra função. Neste 
caso, poderemos verificar se $18.247,02 acumulará para $25.000 daqui a cinco anos a 6,5%. A fórmula 
seguinte de verificação cruzada realmente retorna $25.000: 

=VF(6.5%;5;0;-18247,02;0) 
EXEMPLO 13 

Uma propriedade conduz a um aluguel de $25.000 nos próximos 25 anos. Se eu descontar a 8%, quanto eu 
deverei pagar por ela? Assuma um valor zero após 25 anos e que o aluguel é pago postecipado e 
anualmente. 

Função exigida: VP(taxa, nper, pgto, vf, tipo) 
A fórmula seguinte retorna –$266.869,40: 

=VP(8%;25;25000;0;0) 
Este resultado pode ser verificado usando a função TAXA. Esta fórmula retorna 8.00%: 

=TAXA(25;25000;-266869,40;0;0) 
Tipicamente, pagamentos de real estate são feitos antecipados. Em tal caso, o Exemplo 13 seria modificado 
fazendo o argumento Tipo igual a 1. 

EXEMPLO 14 

Assuma que no Exemplo 13 o aluguel de $25.000 seja recebido em perpetuidade. Se descontarmos a 8%, 
quanto deveria ser pago? 

Isto é um exemplo de problema de desconto que o Excel não pode resolver usando suas funções. O problema 
é que não podemos usar “perpetuidade” como o argumento nper. A solução é usar um período de tempo 
muito longo, tal como 1.000 anos. O resultado é certamente preciso o suficiente para a maioria dos 
propósitos. 

Função exigida: VP(taxa; nper; pgto; vf; tipo) 
A fórmula seguinte retorna –$312.500,00: 

=VP(8%;1000;25000;0;0) 
Outra opão é usar uma fórmula para calcular o valor presente: 

VP = PGTO/TAXA 
Para este exemplo, a fórmula seguinte retorna $312.500,00: 

=25000/0,08  
Note que o sinal é diferente porque a fórmula não adotou estritamente a convenção de sinais. 
Se o aluguel é pago antecipado, nós meramente adaptamos isto abordagem “enganando” e usando 1 para o 
argumento Tipo. A fórmula seguinte retorna $337.500,00: 

=VP(8%;1000;25000;0;1) 
A abordagem da fórmula é variada, e a fórmula geral para avaliar lucros antecipados é como segue: 

VP = PGTO*(1+TAXA)/TAXA 

Para este exemplo, a fórmula seguinte retorna $337.500,00: 

=25000*(1+,08)/,08 
Outros exemplos podem ser expressos em termos de desconto, mas os cobriremos mais tarde na seção 
“Problemas de Capitalização Simples”. 

EXEMPLO 15 

Uma propriedade vale atualmente $2.000.000 e está submetida a um aluguel por cinco anos. O comprador 
pagou $1.750.000 por ela. Assumindo nenhum crescimento futuro no valor, qual foi a taxa de desconto? 
 
Função exigida: TAXA(nper; pgto; vp;vf; tipo;estimativa) 
A fórmula seguinte retorna 2,706609%: 

=TAXA(5;0;-1750000;2000000;0) 
O pagamento hoje representa um valor presente negativo. O valor em cinco anos é um direito (positivo) a 
receber. 
Para verificar a resposta, use esta fórmula (que retorna $2.000.000,03): 

=VF(2,706609%;5;-1750000;0) 
O erro de arredondamento é causado pela fixação da taxa a somente seis casas decimais. 
Normalmente, o argumento seria uma célula de referência, não um valor fixado. 

EXEMPLO 16 

Um arrendatário se interessou por uma propriedade que foi recentemente vendida por $230.000. O 
arrendamento teve quatro anos de duração, e custou $6.000 por mês antecipados sem revisão de 
arrendamento ou escalonamento. Se aceitarmos uma rentabilidade de 0,75%, qual o lucro de aluguel (profit 
rent) que foi apresentado pela transação? Lucro de aluguel é o valor do aluguel menos o arrendamento 
pago. 
 
Função exigida: PGTO(taxa; nper; vp;vf; tipo) 
A fórmula seguinte retorna $5.680,95 (lucro de aluguel): 

=PGTO(0,75%;48;-230000;0;1) 
Adicionando ao arrendamento pago ($6.000) produz um valor de aluguel de $11.680,95. 

Problemas Complexos de Desconto  

Problemas complexos de desconto envolvem o uso de três quantias monetárias: valor presente, pagamento e 
valor futuro. Os exemplos de desconto complexo nesta seção são essencialmente reformulações dos 
problemas de capitalização complexos. 
 

Todos os exemplos nesta seção estão disponíveis nas Planilhas da Pasta Funções 
Financeiras do Excel. 
EXEMPLO 17 

Se eu desconto a 0,75% por mês, quanto deverei pagar por uma propriedade rendendo $25.000 por mês 
antecipados (a qual estimo que valerá $5.000.000 daqui a cinco anos)? 

Função exigida: VP(taxa; nper; pgto; vf; tipo) 
A fórmula seguinte retorna –$4.406.865,34: 

=VP(0.75%;60;25000;5000000;1) 
Este exemplo usa a taxa por mês, e os pagamentos são mensais. Portanto, o argumento nper foi convertido 
para meses. 

Podemos verificar este cálculo usando a função TAXA. A fórmula seguinte retorna 0,75%: 

=TAXA(60;25000;-4406865,34;5000000;1) 
EXEMPLO 18 

Eu paguei $1.200.000 por uma propriedade que rende um aluguel de $12.000 por mês antecipados. Se eu 
vendê-la daqui a  cinco anos por $1.500.000, que lucro receberei? 

Função exigida: TAXA(nper; pgto; vp; vf; tipo; estimativa) 
A fórmula seguinte retorna 1,29136%: 

=TAXA(60;12000;-1200000;1500000;1) 
Este resultado pode ser verificado usando a função VP. A fórmula seguinte retorna –$1.200.000,00: 

=VP(1,29136%;60,12000;1500000;1) 
É importante entender que o aluguel é cotado mensalmente e antecipados, mas o prazo são cinco anos. Esta 
discrepância é resolvida convertendo os anos para meses. 

Portanto, a fórmula retorna uma taxa de juros mensais. 

Note que o aluguel não é convertido para um aluguel anual.  Isto porque um aluguel de 
$12.000 por mês antecipados não é o mesmo que um aluguel de $144.000 por ano antecipados. 
Para conseguir a quantia anual equivalente precisaríamos conhecer a taxa de desconto — que é 
um poucoda informação que estamos tentando calcular. 
EXEMPLO 19 

Uma propriedade foi comprada por $1.600.000. Ela rende um aluguel de $10.000 por mês antecipados. Se 
eu estou seguro de uma renda de 1% por mês, quanto deve valer a propriedade daqui a cinco anos quando 
eu planejo vendê-la? 

Função exigida: VF(taxa; nper;pgto;vp;tipo) 
Esta fórmula retorna $2.081.851,05: 

=VF(1%;60;10000;-1600000;1) 
Este resultado pode ser verificado usando a seguinte fórmula (que retorna –$1.600.000): 

=VP(1%;60;10000;2081851,05;1) 

Problemas de Amortização  
A Amortização é um termo dado ao processo de pagar de volta empréstimos. Neste capítulo, de fato, já foi 
coberto a maioria dos cálculos exigidos, mas os problemas foram expressos em termos de capitalização. 
 

Todos os exemplos básicos desta seção estão disponíveis nas Planilhas da Pasta Funções 
Financeiras do Excel. 
EXEMPLO 20 

Quais são os pagamentos sobre um empréstimo de $4.000.000 durante 20 anos, a 0,75% de juros ao mês 
(com pagamentos postecipados)? 

Este exemplo está ilustrado na Figura 1-4. 
 
 

 
 
Figura 1- 4: Calculando um pagamento de empréstimo 
Função exigida: PGTO (taxa;nper;vp;vf; tipo) 
A fórmula seguinte retorna $35.989,04: 

=PGTO(0,75%;240;4000000;0;0) 
Este resultado pode ser verificado usando a função VP para calcular a quantia emprestada. A fórmula 
seguinte retorna $200.000: 

=VP(0.75%;240;-35989,04;0;0) 
Neste exemplo, o empréstimo é completamente reembolsado após 20 anos, e o argumento vf é zero. 
Também note que os pagamentos serão mensais, e a taxa de empréstimo mensal foi cotada. Portanto, o prazo 
de 20-anos é convertido para meses. 

EXEMPLO 21 

Eu posso me permitir pagamentos de $2.500 por mês, e posso tomar emprestado a 0,45% (por mês) durante 
20 anos. Quanto eu posso me permitir tomar emprestado em uma hipoteca completamente resgatável? 

Função exigida: VP(taxa; nper;pgto;vf;tipo) 
Esta fórmula retorna $366.433,74: 

=VP(0,45%;240;-2500;0;0) 

Note que, com hipotecas, sempre assumimos que os pagamentos sejam postecipados e que o argumento tipo 
seja 0. Note também que a taxa de juros (e os pagamentos) são mensais. Portanto, o prazo de 20 anos deve 
ser convertido para meses. 

Você pode verificar a resposta usando a resposta calculada para determinar a taxa sobre uma hipoteca de 
$366.433,74 durante 240 meses. A fórmula seguinte retorna 0,45%: 

=TAXA(240;-2500;366433,74;0;0) 
EXEMPLO 22 

Eu devo atualmente $150.000 numa hipoteca, e faço pagamentos de $1.900 por mês. A taxa de juros 
corrente é 0,45% por mês. Quanto tempo levará para reembolsar o empréstimo? 

Função exigida: NPER(taxa;pgto;vp;vf;tipo) 
A fórmula seguinte retorna 97,76: 

=NPER(0,45%;-1900;150000;0;0) 
Devido aos juros e pagamentos serem mensais, a fórmula retorna um período de amortização em meses. Esta 
resposta, embora correta em termos matemáticos, tem uma implicação prática. Os pagamentos são realmente 
feitos exatamente nos aniversários mensais. Este cálculo implica que o empréstimo de certo modo fica 
reembolsado no 0,76 do caminho do 98º mês. Na realidade, você tem uma escolha: fazer um pagamento 
adicional no final de 97 meses, ou fazer um pagamento de nível reduzido após 98 meses. Estas opções 
podem ser calculadas usando a função VF. 

Para calcular o pagamento adicional no final dos 97 meses, calcule a quantia devida usando esta fórmula (a 
qual retorna –$1.429,85): 

=VF(0,45%;97;-1900;150000;0) 
Portanto, o final pagamento após 97 meses é –$3.329,85 (isto é, o pagamento normal de –$1.900 mais –
$1.429,85). 
Para calcular o pagamento reduzido após 98 meses, use esta fórmula (que retorna +$463,72): 

=VF(0,45%;98;-1900;150000;0) 
Portanto, o pagamento final após 98 meses é –$1.436,28 (isto é, o pagamento normal de –$1.900 mais 
$463,72). 

Um problema relativamente freqüente surge quando o pagamento for menor que a quantia 
da porção de juros do saldo devedor. Neste exemplo, o outstanding empréstimo é $150.000, e os 
juros do primeiro mês são $675 ($150.000 * 0,45%). Se o pagamento for menor que esta quantia, 
o saldo devedor continuará a aumentar, e o empréstimo se estenderá ao infinito (ou melhor, 
parece sobreviver até o infinito). Se isto acontecer, a função NPER retorna a mensagem de erro 
#NUM!. 
EXEMPLO 23 

Um contrato de crédito ao consumidor fornece que eu tomo emprestado $1.000 e pago $100 por mês 
antecipados por 12 meses. Qual é a  taxa de juros? 

Função exigida: TAXA(nper;pgto;vp;vf;tipo;estimativa) 
A fórmula seguinte retorna 3,503153%: 

=TAXA(12;-100;1000;0;1) 
Antes de você começar a pensar quão generoso é este contrato, lembre-se que os pagamentos são por mês. 
Portanto, o resultado é a taxa efetiva mensal! 

A taxa efetiva anual equivalente é 51,16%, calculada como segue: 

=((1+0,03503153)^12)-1 
A taxa anual, baseada na nominal composta com base mensal, retorna 42,05%, calculado como segue: 

=3,503153 * 12 

Há uma grande diferença entre a taxa efetiva anual e a taxa nominal equivalente composta 
mensalmente. O tamanho da diferença aumenta com os níveis das taxas usados. 
EXEMPLO 24 

Eu emprestei $300.000 de uma hipoteca “balloon” durante 15 anos, com pagamentos mensais de $100.000. 
O saldo de $200.000 é devido no final do prazo. A taxa de juros é 0,4% por mês, e os pagamentos são feitos 
mensalmente e postecipados. Quais serão os pagamentos? 

Um tipo de hipoteca comum (usado para aumentar a quantia que eu posso tomar emprestado) é a assim 
chamada hipoteca “balloon”. O empréstimo é dividido em dois elementos: 1) o elemento “pagamento”, onde 
os pagamentos cumprem compleatmente parte do empréstimo no final do prazo, e 2) o elemento “balloon”. 
Durante o prazo do empréstimo, somente os juros (e não o principal) são pagos no elemento balloon. O saldo 
principal é pago como um montante no final do empréstimo. 

A habilidade para uso de um argumento vf nas funções VP, PGTO, TAXA, e NPER torna-se relativamente 
fácil de realizar nos cálculos de hipoteca balloon. 

Função exigida: PGTO(taxa;nper; vp; vf;tipo) 
A fórmula seguinte retorna –$1.580,41: 

=PGTO(0,4%;180;300000;-200000;0) 
Note que a hipoteca total de $300.000 é usada para o argumento vp. 
Este cálculo pode ser verificado usando o pagamento calculado para determinar o VP. Esta fórmula retorna 
$299.999,43 (o erro de arredondamento foi causado pelo uso de uma quantia de pagamento arredondada): 

=VP(0,4%;180;-1580,41;-200000;0) 
Os pagamentos com base balloon podem ser comparados com pagamentos de uma hipoteca tradicional. Esta 
fórmula retorna $202.509,64 (hipoteca tradicional): 

=VP(0,4%;180;-1580,41;0;0) 
E pagamentos para a hipoteca tradicional de $300.000 são –$2.341,24, calculado com esta fórmula: 

=PGTO(0,4%;180;300000;0;0) 
Os exemplos de cálculos de amortização anteriores podem ser modificados para as hipotecas balloon 
Fornecendo um argumento vf nas funções VP, PGTO, NPER e TAXA. 

Você pode calcular também o elemento hipoteca balloon por si só com a função VF. 

Isto é um cálculo que exige uma interpretação cuidadosa do sinal do resultado. Se a função VF retornar um 
valor positivo, que significa que a hipoteca original foi paga a mais e esta quantia é agora devida ao tomador 
do empréstimo. Se ela retornar uma quantia negativa, esta é a quantia do elemento balloon. Um elemento 
balloon existirá nos casos onde as quantias de pagamentos não pagam o empréstimo completamente durante 
o prazo da hipoteca à taxa de juros cotada. 

Tipicamente, estes cálculos são feitos em dois estágios. Primeiro, calcule o pagamento sobre a amortização 
normal do empréstimo (usualmente de acoro com as regras do credor). Segundo, calcule quanto o elemento 
“balloon” num pagamento adicional permitirá. O Exemplo 25 fornece os detalhes. 

EXEMPLO 25 

Se o banco insiste numa amortização de $200.000 de um empréstimo, quanto extra eu posso tomar 
emprestado sobre uma hipoteca balloon se eu posso me permitir pagamentos de $3.000 por mês? O prazo 
do empréstimo é 10 anos, e a taxa atual é 0,4% por mês. 
 
Função exigida: PGTO(taxa; nper; vp; vf; tipo) 

O primeiro passo é calcular o pagamento para uma amortização normal do empréstimo de $200.000. A 
fórmula seguinte retorna –$2.101,81: 

=PGTO(0,4%;120;200000;0;0) 
Se os pagamentos de $3.000 forem a preços acessíveis, a quantia adicional de $898,19 pode ser paga como 
juros sobre o elemento balloon (isto é, $3.000 – $2.101,81). O elemento balloon pode agora ser calculado 
porque a quantidade de juros é conhecida. Esta fórmula, que representa o elemento balloon, retorna 
$224.546,88: 

=898,19 / 0,4% 
O cálculo pode ser verificado calculando o pagamento baseado numa hipoteca total de $424.546,88 com um 
elemento balloon de $224.546,88. A fórmula seguinte retorna –$3.000: 

=PGTO(0,4%;120;424546,88;-224546,88;0) 

Convertendo Taxa de Juros  
Os exemplos anteriores foram convenientemente expressos para permitir uma adequação fácil das taxas de 
juros com a freqüência de pagamentos e o prazo total. Freqüentemente, entretanto, interpretar um problema 
financeiro será mais difícil. Existem duas situações nas quais a conversão de taxa de juros deve ser feita: 

_ Quando você deve fazer os cálculos envolvendo a freqüência de pagamentos ou a número de períodos, e a 
taxa que você exige para usar não se adapta à freqüência de pagamentos ou período. 

_ Quando você tiver feito cálculos envolvendo a freqüência de pagamentos ou a número de períodos, e você 
precisa expressar a taxa de juros resultante em termos de uma taxa por ano ou algum outro período de 
tempo. 

Para criar fórmulas precisas, você precisará entender o princípio de equivalência das taxas de juros. 
Colocando simplesmente, qualquer taxa de juros por um período de tempo é equivalente a outra taxa de 
juros por um diferente período de tempo. 

Métodos de Cotizar a Taxa de Juros 

Existem três métodos usados comumente para cotização das taxas de juros: 

_ Taxa Nominal: O juro é cotado com base anual, junto com uma freqüência de composição por ano. Por 
exemplo, a APR cotada de, digamos, 6% composto mensalmente, onde 0.5% é devido por mês. 

 
_ Taxa efetiva anual: A taxa de juros na qual a taxa dada representa a porcentagem ganha em um ano. Por 
exemplo, com uma taxa efetiva anual de 10%, $1.000 rende $100 de juros no final de um ano. 

_ Taxa efetiva periódica: A taxa de juros em que a taxa dada representa a porcentagem ganha durante um 
período de menos que um ano. Por exemplo, com uma taxa de 3% por semestre, $300 rende $9 após seis 
meses. 

Uma taxa de juros cotada usando qualquer um destes três métodos pode ser convertida a qualquer outra 
pelos três métodos. Por exemplo, considere uma taxa de juros de 1% por mês sobre $100. No primeiro mês, 
o investmento ganha, $1 de juro. Se os juros creditados não fôr sacado, ele será adicionado ao principal, e o 
juro subseqüente será baseado no novo saldo. Uma taxa de juros mensal de 1% é equivalente a 12,6825% 
por ano de taxa de juros (a taxa efetiva). Isto é calculado usando a seguinte fórmula: 

=(1+0.01)^12 – 1 
Outro exemplo de uma Taxa Nominal é uma taxa de juros cotada como 6% por ano, composta 
trimestralmente. Isto significa que 1,5% (isto é, 6% / 4) é pago ou recebido a cada três meses. 

A maioria dos bancos e instituições financeiras cota os juros com base nominal composto mensalmente. 
Entretanto, quando se informa o retorno de investmentos ou quando se compara taxas de juros, é comum 
cotar o retorno efetivo anual, que torna mais fácil para comparar as taxas. Por exemplo, sabemos que 12% 
por ano composto mensalmente é mais que 12% por ano composto trimestralmente — mas não sabemos 
(sem um cálculo de conversão intermediário) quanto mais ele é. 

Convertendo Taxa de Juros Usando o Suplemento de Funções Financeiras  
Como você verá, 10 conversões diferentes podem ser exigidas na conversão entre os sistemas Nominal, 
Efetivo Anual e Efetivo Periódico. 

As planilhas de Funções Financeiras no Excel que estão anexadas contêm um suplemento 
(chamado Funções Financeiras), escrito por Norman Harker. Este suplemento fornece as funções 
personalizadas (escritas em VBA) para calcular as conversões de taxas de juros. Você encontrará 
também uma pasta que demonstra o uso destas funções. Além disso, estas funções são usadas 
em muitos dos exemplos neste e nos capítulos subseqüentes. Para a sua conveniência, as 
funções VBA são definidas na pasta exemplo. Portanto, você não precisa instalar o suplemento 
para trabalhar com as pastas exemplos. 
 
Quando usar o suplemento das Funções Financeiras, você pode ou entrar com a função manualmente, ou 
usar a caixa de diálogo Inserir Função do Excel (as funções estão localizadas na categoria Financeiras).  
O nome da função e os argumentos podem parecer confusos à primeira vista, mas você logo os entenderá. O 
nome de cada função é composto de três partes: 

_ A taxa de juros você tem (Effx, AnnEff, ou Nomx). Note que a freqüência de composição da Taxa Efetiva 
e Nominal são denotadas por x. 

_ O símbolo linking, que é um caractere sublinhado (_). 

_ A taxa de juros você quer (Effx, Effy, AnnEff, Nomx, ou Nomy). Novamente, as freqüências de 
composição são denotadas por x (se ela for a mesma que a freqüência da taxa que você tem), ou y (se ela for 
diferente). 
A ordem dos argumentos também é fácil de controlar: 

_ O primeiro argumento é sempre a taxa de juros que você tem. 

_ O segundo argumento é sempre a Freqx, que é a freqüência da taxa Effx ou Nomx. Note que cada função 
de conversão usa um argumento Freqx, e é sempre o segundo argumento. 

Se existir uma segunda freqüência conhecida diferente de x ou anual, há um terceiro argumento, Freqy. 

Custo Efetivo de Empréstimos  



Bertolo  Introdução às Funções Financeiras do Excel   

Página 20                                                                                                                                                              Finanças no Excel    

Instituições de empréstimos tipicamente publicam em suas “manchetes” as taxas para fazê-las parecer tão 
baixa quanto possível. Um tomador do empréstimo experiente é capaz de interpretar estas taxas para 
determinar quanto o empréstimo está custando realmente. A única comparação segura e constante é observar 
o custo efetivo em termos das taxas de juros efetivas anuais, ou alguma outra taxa comum tal como a taxa 
nominal anual composta mensalmente. 

Esta seção apresenta quatro exemplos que demonstram como calcular o custo efetivo dos empréstimos. 

Todos os exemplos desta seção estão disponíveis nas Planilhas da Pasta Funções 
Financeiras do Excel. Estes exemplos usam as funções de conversão de taxas de juros 
personalizadas do VBA. 

Impacto de Fees e Charges sobre o Juro Efetivo 

Além disso, para o interessado sobre uma hipoteca, os bancos cobram freqüentemente “pontos,” ou set-up 
fees, e taxas de serviços de conta. Estas taxas adicionam-se ao custo efetivo do empréstimo. Mas por 
quanto? 

EXEMPLO 26 

Um banco cota uma hipoteca taxa de 3% nominal composta mensalmente, e você está interessado em tomar 
um empréstimo de R$300.000 durante 20 anos com pagamentos mensais. O banco cobra de antemão ao 
acordo do empréstimo uma taxa de 1% do empréstimo, mais uma taxa de serviço da conta de $15 por mês. 
Qual é o custo efetivo anual do empréstimo? 

A Figura 1-5 mostra uma planilha que é montada para resolver este problema. A informação conhecida é 
entrada na seção base de Dados da planilha. A Tabela 1-2 lista as fórmulas centrais que realizam os cálculos. 
Para clareza, as fórmulas são mostradas usando valores reais ao invés de células referências. 

A Célula B19 usa a função EFETIVA(Nominal/12;12). 
 
Os pagamentos são baseados na quantia de empréstimo de $150.000, mas o custo efetivo está baseado no 
fato que, após deduzir o set-up fee, o tomador do empréstimo recebe somente $147.000. Similarmente, 
actual pagamentos são mais altos pela quantia de taxas de serviços da conta. 

O impacto destes custos varia de acordo com o prazo: Quanto meno for o prazo, maior o impacto. Se uma 
hipoteca não for capaz de ser transferida para uma nova casa quando o tomador do empréstimo se mudar, o 
cálculo deverá ser baseado no tempo provável que a hipoteca encerrará — usualmente cerca de sete anos. 

Empréstimos de Taxa “Flat” 

Muitos contratos de crédito ao consumidor usam um contrato de empréstimo nos quais uma porcentagem do 
empréstimo é adicionada ao empréstimo, e os pagamentos são baseados no agregado da quantia emprestada 
mais o juro flat dividido pelo número de pagamentos. Podemos usar as funções EFETIVA e TAXA do Excel 
para calcular os custos efetivos de tal empréstimo. 

EXEMPLO 27 

Um consumidor financia a compra do seu carro com uma taxa empréstimo flat de $15.000 durante 18 
meses. Juros de 10% * 1,5 esta quantia é adicionado ao empréstimo e ele paga 1/18 desta quantia cada mês 
antecipados por 18 meses. Qual é o custo efetivo do empréstimo? 

A maneira mais fácil para resolver esta função é usar a função EFETIVA. A fórmula seguinte retorna 3,62%: 

=EFETIVA(TAXA(18;-17250/18;15000;0;1);12) 
Note que se o prazo de tal empréstimo for somente 12 meses, a taxa é ligeiramente mais do que o dobro da 
taxa flat. Muitos estados e nações têm legislação para tais contratos de empréstimo teriam a Taxa Anual 
Nominal composta mensalmente estabelecido claramente no contrato de empréstimo. 

Empréstimos Livres de Juros 

Outro cálculo interessante é o custo efetivo de uma assim chamada oferta de empréstimo “livre de juros”. 
Fazendo estes cálculos, você precisa saber o preço para o qual você conseguirá o produto em outro lugar 
(sem o pacote de livre de juros). 

EXEMPLO 28 

Um consumidor compra um sistema hi-fi ao preço de tabela por $3.000 em termos de “livre de juros” 
durante 12 meses, com os pagamentos antecipados. Ele poderia ter comprado um sistema idêntico por 
$2.500 a vista ou em termos de crédio normal. Qual é o custo efetivo deste empréstimo? 

Novamente, a função Effx_AnnEff VBA fornece a solução mais simples. Esta fórmula retorna 51,16%: 

=Effx_AnnEff(TAXA(12;-(3000/12);2500;0;1);12) 
Tais cálculos são freqüentemente mais difíceis quando o preço a vista equivalente é subjetivo (por exemplo, 
o mercado de veículos usados). 

Você pode fazer cálculos similares para outros tipos de contratos, tais como “Pagar 25% hoje, não pagar 
mais por 12 meses.” Novamente, o segredo é conseguir o preço a vista equivalente, e daí então comparar os 
cálculos com aquele preço, rather than um preço que esteja inflacionado pelo varejista que está oferecendo o 
crédito. 

Muitos estados e nações têm legislação de crédito ao consumidor que governam a cotação das taxas de juros. 
Em muitos locais, o únicio regulamento de contratos do tipo livre de juros é que o varejista não pode 
oferecer o mesmo produto a preço a vista diferente daquele cotado no contrato livre de juros. 

Custos de Empréstimos de “Pagamentos Anuais / 12” 

Uma prática que está enraizada nos dias anteriores à calculadora é calcular pagamentos na base “anual 
postecipados”, e to cobrar the tomador do empréstimo 1/12 daquela quantia cada mês. Este cálculo foi 
facilitado pelas tabelas pré-preparadas de pagamentos mensais por $1.000 de empréstimo. A prática prevails 
(especialmente nas UK Building Societies) parcialmente porque ela produz uma taxa de publicação mais 
baixa que os regimes de Taxa Nominal ou Efetiva. 

EXEMPLO 29 

Um banco oferece uma hipoteca de $100.000 à taxa de7% durante 10 anos, onde os pagamentos por mês 
são baseados em 1/12 do pagamento anualmente calculado sendo pagos mensalmente e postecipados. Qual 
é o custo efetivo anual? 

A fórmula seguinte (que usa a função Effx_AnnEff VBA) retorna ,7522% (a taxa efetiva por ano): 

=Effx_AnnEff(TAXA(10*12;PGTO(7%;10;100000;0;0)/12;100000;0;0);12) 

Calculando os Componentes de Juros e Principal 
Esta seção discute quatro funções do Excel que o habilita a: 

_ Calcular os componentes de juros ou principal de um particular pagamento (as funções IPGTO e PPGTO) 

_ Calcule os componentes acumulados de juros ou principal entre quaisquer dois períodos 

 

Os exemplos nesta seção estão disponíveis nas Planilhas da Pasta Funções Financeiras do 
Excel. 

Usando as Funções IPGTO e PPGTO  

Você poderá querer saber (ou simplesmente estar curioso sobre) quanto de um pagamento particular 
constitui os juros, e quanto do pagamento vai para o principal. 

Esta informação poderá ser útil na determiunação de efeitos de impostos sobre os pagamentos dos juros. Se 
você estudou qualquer um dos exemplos de amortização de empréstimo, você sabe que o elemento juros não 
é constante durante a vida de um empréstimo. Ao contrário, o componente juro decresce, enquanto o 
componente principal cresce. 

Se você criou uma planilha de amortização, estas funções não são particularmente úteis, 
porque você pode simplesmente refer à planilha. As funções IPGTO e PGTO são mais úteis 
quando você precisar determinar o desdobramento de juros/principal de um pagamento particular. 
A sintaxe para estas duas funções é como segue (argumentos em negrito são exigidos): 

IPGTO(taxa,per,nper,vp,vf,tipo) 
PPGTO(taxa,per,nper,vp,vf,tipo) 
Como com todas as funções, a taxa, per, e nper devem coincidir em termos do período. Se o prazo do 
empréstimo for medido em meses, o argumento taxa deve ser a taxa efetiva por mês, e o argumento (isto é, o 
período de juros) deve ser um mês particular. 

EXEMPLO 30  

Um consumidor obteve um empréstimo de carro por três-anos (pagamentos mensais) por $20.000 a uma 
taxa anual de 8%. Quais são as porções de juros e principal para o pagamento final do empréstimo? 

A Figura 11-6 mostra a solução, monte numa planilha. 

Figura 1-6: Esta planilha calcula os componentes juros e principal para quaisquer períodos 
de um empréstimo. 
Função exigida: IPGTO(taxa;per;nper;vp;vf;tipo) 
Esta fórmula calcula a porção de juros do pagamento final, e retorna –$4,15: 

=IPGTO(8%/12;36;36;20000;0;0) 
A fórmula seguinte calcula a porção do principal do pagamento final, e retorna –$622,58: 

=PPGTO(8%/12;36;36;20000;0;0) 
No final do prazo do empréstimo, praticamente todos os pagamentos seguem adiante do principal. 
Para comparar este com o primeiro empréstimo período, mudo o argumento por para 1. 
Após fazer isto, as fórmulas retornam –$133,33 (juros) e –$493,39 (principal). 

Você pode verificar os cálculos usando a função PGTO (que retorna o pagamento total, 
juros mais principal). A fórmula seguinte retorna –$626.73, que é a quantia de pagamento do 
empréstimo (e a soma das duas fórmulas anteriores): 
=PGTO(8%/12;36;-20000;0) 

Usando as Funções PGTOJURACUM e PGTOCAPACUM 

As funções IPGTO e PPGTO podem ser úteis. Mas, muito freqüentemente, você precisará saber os 
componentes juros ou principal para um grupo de períodos consecutivos. Neste caso, as funções 
PGTOJURACUM e PGTOCAPACUM são de grande serviço. Estas funções são úteis para se criar 

planilhas de amortizações anualizadas, e para construir juros qualificados para propósitos de retorno de 
imposto de renda. 

A sintaxe para estas funções é mostrada aqui (todos os argumentos são exigidos): 

PGTOJURACUM(taxa;nper;vp; início _ período; final_ período;tipo_pgto) 
PGTOCAPACUM(taxa;nper;vp; início _ período; final_ período; tipo_pgto) 

Estas funções estão disponíveis somente quando o suplemento Ferramentas de Análise 
estiver instalado. 
EXEMPLO 31 

Um consumidor está tomando emprestado $250.000 numa hipoteca, re-embolsável durante 10 anos a 5,6% 
nominal, composto mensalmente, com pagamentos mensais postecipados. Quais serão os pagamentos dos 
juros e do principal no primeiro ano do empréstimo? 

A fórmula seguinte, para pagamentos do principal, retorna $19.183,06: 

= PGTOCAPACUM (EFETIVA(0,056/12;12);10*12;250000;1;12;0)  
A fórmula seguinte retorna $13.541,55 (pagamento total dos juros): 

= PGTOJURACUM  (EFETIVA(0,056/12;12);10*12;250000;1;12;0) 
Podemos verificar estas respostas usando a função PGTO para calcular o agregado dos pagamentos. A 
fórmula seguinte retorna $32,724.61, que é a soma dos resultados precedentes: 

=PGTO(EFETIVA(0,056/12;12);10*12;250000;0;0)*12 

Estas fórmulas todas usam a função personalizada do VBA Nomx_Effx. 

Compatibilizando Juros e Freqüências de Pagamento 
Diferentes  
Os exemplos anteriores envolveram juros nominais com freqüências de composição que se adaptam às 
freqüências de pagamentos. Assim, por exemplo, poderíamos ter uma taxa nominal composta mensalmente 
cotada com pagamentos que também são mensais. Como usual, o mundo real não é sempre cooperativo. 

EXEMPLO 32 

Um banco cota a taxa nominal composta mensalmente em 6,3%, mas permite pagamentos semanais à taxa 
de juros equivalente. Se eu tomar emprestado $300.000 durante 10 anos, quais seriam os pagamentos 
semanais? 

O modo fácil para resolve tais problemas é usar a função de conversão de juros personalizada Nomx_Effy. 
Esta fórmula retorna $777,51: 

=PGTO(Nomx_Effy(6,3%;12,52);10*52;300000;0;0) 
EXEMPLO 33 

Nós temos montado uma conta anual, mas precisa manipular uma saída mensal de $12.500. Ao invés de 
anualizar multiplicando por 12, qual é a quantia anual equivalente usando um depósito de 7% por ano 
nominal composta mensalmente? O pagamento mensal é postecipado, e a quantia equivalente é para ser 
calculada no final de cada ano. 

Primeiro, calcule a taxa efetiva mensal (usando uma função personalizada do VBA). A fórmula seguinte 
retorna 0,58333%: 

=Nomx_Effx(7%;12) 
Então, calcule a quantia anual equivalente usando a função VF. Esta fórmula retorna –$154.907,29: 

=-VF(0,58333%;12;-12500;0;0) 

Neste exemplo, os sinais podem ser confundidos. Normalmente trataríamos a saída como 
um valor futuro negativo e o retorno como positivo. Entretanto, se estivermos usando o resultado 
como uma saída, então os sinais são revertidos. Isto pode ser feito ou usando – 12.500, como a 
saída, ou invertendo o sinal do resultado usando – VF (como no exemplo). 
Se a quantia equivalente for calculada antecipadamente, usaríamos os mesmos princípios e aplicaríamos a 
função VP. 

Limitações das Funções Financeiras do Excel 
As principais funções financeiras do Excel (VP, VF, PGTO, TAXA, NPER, PGTOJURACUM e 
PGTOCAPACUM) são muito úteis, mas elas têm duas limitações comuns: 

_ Elas podem manipular somente um nível de taxa de juros. 

_ Elas podem manipular somente um nível de pagamento. 

Por exemplo, a função NPER não pode manipular as variações nos pagamentos que surgem com cálculos de 
cartão de crédito. Em tais cálculos, o pagamento mensal é baseado numa redução do saldo devedor, e pode 
estar também sujeito a uma regra de quantia mínima. 

A solução comum para o problema de pagamentos variável é criar uma planilha de fluxo de caixa e usar 
outras funções financeiras que possam manipular pagamentos múltiplos e taxas. Exemplos do processo 
aparecem nos próximos dois capítulos. Brevemente, as funções envolvidas são: 

_ VFPLANO, que trata da capitalização de um Valor Presente em diferentes taxas e que, quando usada 
numa fórmula, pode calcular o valor presente de uma quantia futura em diferentes taxas. 

_ TIR, que trata do cálculo de uma taxa única de fluxos de caixa regulares. 

_ VPL, cálculo da soma dos valores presentes de fluxos de caixa regulares e que pela fórmula pode 
manipular a soma de valores acumulados de fluxos de caixa regulares. 

_ MTIR, que é uma TIR especialista voltada para evitar problemas de TIR múltiplas aplicando diferentes 
taxas aos fluxos de caixa regulares negativos e positivos. 

_ XTIR, que trata do cálculo de uma única taxa de fluxos de caixa irregulares. 

_ XVPL, que trata do cálculo da soma dos valores presentes de fluxos de caixa irregulares e que, numa 
fórmula, pode tratar a soma dos valores acumulados de fluxos de caixa irregulares. 

Numa situação que envolve somente uma ou duas variações, pode ser possível evitar a construção de fluxo 
de caixa usando fórmulas aninhadas nela, ou aplicada às fórmulas básicas de amortização. 
 

Início Diferido para uma Série de Pagamentos Regulares 

Em alguns casos, uma série de fluxos de caixa pode ter um início diferido. Podemos calculcar o VP de uma 
série regular de fluxos de caixa com um início diferido usando uma fórmula como esta: 

=VP(TAXA;NPER;PGTO;VF;Tipo)*(1+TAXA)^-DEFER_PER 
Aqui, DEFER_PER representa o número de períodos para os quais o primeiro fluxo de caixa é diferido. 

EXEMPLO 34 

Eu quero tomar emprestado dinheiro com base em pagamentos diferidos. O período de carência será um 
ano. Depois disso, o empréstimo será por 10 anos com pagamentos mensais postecipados. A taxa de juros é 
de 8% por ano e efetiva. O empréstimo é para adquirir uma propriedade que eu estou construindo, e o 
banco está oferecendo o empréstimo, sujeito a pagamentos não excedendo a 75% do lucro estimado de 
$9.500 por mês. Quanto eu devo tomar emprestado? 

A fórmula seguinte usa a função personalizada AnNEff_Effx, e retorna $550.422,02: 

=VP(AnnEff_Effx(8%,12,10*12,-9500*75%,0,0)*(1+AnnEff_Effx(8%,12))^- 
12 

Avaliando uma Série de Pagamentos Regulares 

Podemos estender o princípio básico de desconto, mas diferentes, níveis de pagamentos encadeando as 
funções VP. Por exemplo, se PV1, PV2 e PV3 representam diferentes valores presentes de séries de 
pagamentos por períodos NPER1, NPER2 e NPER3, o valor descontado de todas as séries de pagamentos 
podem ser encontrados por: 

PV1 + PV2(1+I)^-NPER1 + PV2(1+I)^-(NPER1+NPER2) 
EXEMPLO 35 

Qual é o valor presente de uma propriedade rendendo $5.000 por mês por quatro anos, aumentando para 
$6.500 por mês nos próximos três anos, e aumentando para $8.500 por mês nos três anos finais? Após 10 
anos, a propriedade terá um valor estimado de $1.300.000. Uma taxa de desconto de 10% por ano pode ser 
assumida e todos os pagamentos são antecipados. 

A fórmula seguinte retorna –$978.224,54: 

=VP(AnnEff_Effx(10%,12),48,5000,0,1) + 
VP(AnnEff_Effx(10%,12),36,6500,0,1)*(1+AnnEff_Effx(10%,12))^-48 + 
VP(AnnEff_Effx(10%,12),36,8500,1300000,1)*(1+AnnEff_Effx(10%,12))^-(48+36) 
 
Note como o valor final de $1.300.000 foi aninhado na função VP final. 

A mesma resposta poderia ser obtida pelo “aninhamento” de successivos Valores Presentes dentro da função 
precedente como valores futuros. Mas lembrando que como o VP naquele momento representa um direito à 
série de lucros futuros, o sinal teria que ser revertido. A fórmula seguinte retorna $978.224,54: 

=VP(AnnEff_Effx(10%,12),48,5000,- 
VP(AnnEff_Effx(10%,12),36,6500,- 
VP(AnnEff_Effx(10%,12),36,8500,1300000,1),1),1) 
Destas duas abordagens, a primeira fórmula (usando as fórmulas básicas de desconto) parece mais fácil 
como um método; ela parece mais fácil para construir usando a técnica da megafórmula ou desdobrar em 
três células que são adicionadas juntas. 

A fórmula seguinte retorna $200.344,00: 

=VP(AnnEff_Effx(10%,12),48,5000,0,1) 
Esta fórmula retorna $139.559,07: 

=VP(AnnEff_Effx(10%,12),36,6500,0,1)*(1+AnnEff_Effx(10%,12))^-48 
Esta fórmula retorna $638.331,47: 

=VP(AnnEff_Effx(10%,12),36,8500,1300000,1)*(1+AnnEff_Effx(10%,12))^- 
(48+36) 
E o total dos três elementos checks at $978.224,54. 

Sujeito as exceções envolvendo apenas uma ou duas variações nas séries de pagamentos, a solução será 
montar uma planilha de fluxos de caixa. Isto será coberto após o próximo capítulo porque teremos que 
primeiro descrever as ferramentas básicas VPL e TIR. 

Sumário 
Este capítulo introduziu as funções financeiras e forneceu os conceitos básicos do valor do dinheiro no 
tempo e taxas de juros equivalentes. O capítulo apresentou uma série de exemplos que usaram as funções 
financeiras principais para capitalização, desconto e amortização de empréstimos. 

O próximo capítulo apresenta exemplos que usam o Excel para cálculos de depreciação, e introduz as 
técnicas de cálculo do valor presente líquido (VPL) e taxas internas de retorno (TIR). 

 

 

 


